APPENDIX I

```
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;                              TRANSMITTER CODE
;
;                      COPYRIGHT © POINT SIX WIRELESS, LLC 2002
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

base $10

 INCLUDE 'H705J1A.asm'                ; equates for the HC705J1A

; This version is the COUNTER IR interface
; The clock taps the main processor 16/second.
; This version has microprocessor power level selection via pullups.


                org     MOR
                fcb     $A0          ; SHORT OSC DELAY

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;
;     Equates and RAM Storage
;
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;
;     I/O Pin Equates:
;
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

PWRLEVL       equu     0                 ; PA0, POWER LEVEL LOW
CONTROL
PWRLEVM       equ      1                 ; PA0, POWER LEVEL MEDIA
CONTROL
PWRLEVH       equ      2                 ; PA0, POWER LEVEL HIGH
CONTROL
IRDOUT        equ      3                 ; PA3, ir data, output
LED           equ      5                 ; PA5, LED, OUTPUT
RBASE         equ      7                 ; PA7, RESET BASE TIMMING,
OUTPUT

PUHBUT        EQU      2                 ; pb2, PUSH BUTTON INPUT


        org RAM

QUIETMO       RMB      1                 ; QUIET MODE FLAG
PBCTR         RMB      1                 ; PUSH BUTTON DOWN TIMER
HIPOWER       RMB      1                 ; HIGH POWER MODE FLAG
```

```
PSTIMER       RMB      1                    ; POWER SELECT TIMER


              org      EPROM                ; start at the top of EPROM

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

POWER         RSP
              lda    #%01000111             ; set IR output for active
state
              sta    PORTA                  ; release ir data
              lda    #%11110100             ; set IR data output so that
                                            ; IR is off
              sta    ddra                   ; HIGH POWER MODE
              lda    #%01111111             ; disable pulldowns 0-6
              sta    pdra
              TST    QUIETMO                ; IN QUIET MODE?
              BNE    POWER3

              TST    HIPOWER                ; IN HIGH POWER MODE
              BNE    POWER0
              BSET   PWRLEVL,DDRA           ; DISABLE HIGH POWER CONTROL
              BSET   PWRLEVH,DDRA

POWER0        lda    #$07                   ; 27 microsec 6*7+5=47
              bclr   irdout,porta
              bset   irdout,ddra            ;[5]
POWER1        deca                          ;[3] 1-CYC=.5747mS.
              bne    POWER1                 ;[3]
              bclr   irdout,ddra            ;[5] ir led OFF

              lda    #$0F                   ;[2] 50 microsec 6*15+2=90
POWER2        deca                          ;[3] 1-CYC=.559mS.
              bne    POWER2                 ;[3]
              bset   RBASE,porta            ; reset the base time

POWER3        BSR    PROCPB                 ; PROCESS THE PISHBUTTON

POWER4        TST    PSTIMER                ; LED TIMER ACTIVE
              BEQ    POWER5
              BSET   LED,PORTA
              BSET   LED,DDRA
              DEC    PSTIMER
POWER5              stop

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

PROCPB        BRCLR PUHBUT,PORTB,PBNOT      ; IS THE PB PUSHED?
              TST    QUIETMO                ; FLASH LED IF IN QUIET MODE
              BEQ    PROCPB1

              BSET   LED,PORTA              ; LED ON
              BSET   LED,DDRA
              lda    #$FF                   ;[2] 256*18=4608 cyc
PROCPBD       deca                          ;[3]
```

```
            inca                        ;[3]
            deca                        ;[3]
            inca                        ;[3]
            deca                        ;[3]
            bne   PROCPBD               ;[3]
            BCLR  LED,PORTA             ; LED OFF


PROCPB1     LDA   PBCTR                 ; DO NOT GO ABOVE 255
            CMP   #$FF
            BEQ   POWOOS
            INC   PBCTR                 ; BUMP THE PB ACTIVE COUNTER
            LDA   PBCTR                 ; BETWEEN 4 AND 6 SECONDS
            CMP   #$40                  ; LED ON AT 4 SECONDS
            BLO   POWOOS
            LDA   #$50                  ; set LED timer
            STA   PSTIMER
            STA   HIPOWER               ; SET HIGH POWER MODE
            CLR   QUIETMO               ; EXIT QUIET MODE
            BSET  LED,PORTA             ; LED ON
            BSET  LED,DDRA
            LDA   PBCTR                 ; LED OFF AT 6 SECONDS
            CMP   #$60
            BLO   POWOOS
            BCLR  LED,PORTA
            CLR   PSTIMER               ; NO LED TIME IN LOW POWER
MODE
            CLR   HIPOWER               ; SET LOW POWER MODE
POWOOS      RTS

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

PBNOT       LDA   PBCTR                 ; ENTRY INTO QUIET MODE?
            CMP   #$40                  ; >2 SECONDS, EXIT/ENTER
QUIET MODE
            BLO   PBNOT1
            CMP   #$A0                  ; >10 SECONDS DOWN?
            BLO   PBNOT2
            LDA   #1                    ; SET QUIET MODE
            STA   QUIETMO
PBNOT1      CLR   PBCTR                 ; CLEAR THE PB DOWN COUNTER
            RTS
PBNOT2      CLR   QUIETMO               ; EXIT QUIET MODE
            BRA   PBNOT1

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;;; reset vectors:

            org     $07f8
            fdb     POWER
            fdb     POWER
            fdb     POWER
            fdb     POWER
            end
```

APPENDIX II

```
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;                              RECEIVER CODE
;
;                    COPYRIGHT © POINT SIX WIRELESS, LLC 2002
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

 base $10

 INCLUDE 'H705J1A.asm'              ; equates for the HC705J1A

; This is the Standard version seed = $0000
; This version is the COUNTER IR receiver/418 transmitter
; The clock taps the main processor 16/second.
; The xtal is 4 Mhz.


ID            EQU   $11                  ; COUNTER ACCUMULATOR
SID           EQU   $10                  ; SERVICE ID
XMPERIOD      EQU   $0A                  ; transmitt period = 10 seconds
SEEDMSB       EQU   $00                  ; MSBYTE SEED FOR POINT SIX
                                         ; STANDARD
SEEDLSB       EQU   $00                  ; LSBYTE SEED FOR POINT SIX
                                         ; STANDARD



                org     MOR
                fcb     $A0              ; SHORT OSC DELAY

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;
;        Equates and RAM Storage
;
;
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;
;
; I/O Pin Equates:
;
;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

TPS             equ     $10              ; 16 ticks/second

IRPWR           equ     0                ; PA0, IRPOWER, output
LCDCLR          equ     1                ; PA1, LCD CLR, OUTPUT
LED             equ     2                ; PA2, LED DRIVE ACTIVVE HIGH,
                                         ; OUTPUT
```

```
IRIN            equ     3               ; PA3, ir data, INPUT
LCDCTR          equ     4               ; PA4, LCD CTR, output
SPARE5          equ     5               ; PA5, SPARE, OUTPUT
radio           equ     6               ; PA6, radio transmit data, OUTPUT
RBASE           equ     7               ; PA7, RESET BASE TIMER, OUTPUT

PUHBUT          equ     2               ; PB2, PUSHBUTTON INPUT
SPAREB3         equ     3               ; PB3, SPARE, OUTPUT

OSWITCH         equ     0               ; open-door status bit in STATUS
CSWITCH         equ     1               ; close-door status bit in STATUS


        org RAM

BYCT            RMB     2               ; BYTE COUNT
CHAR            RMB     1               ; CHARACTER STORE FOR SERIAL DATA
TEMPA           RMB     2               ; TEMP STORAGE
SECOND          RMB     1               ; second sub counter
SUBCTR          RMB     1               ; SUB COUNTER
STATUS          RMB     1               ; STATUS BITS 0,1
PBCTR           RMB     1               ; PUSH BUTTON DOWN COUNTER
SETUPMOD        RMB     1               ; TEST MODE FLAG
SETUPTIM        RMB     1               ; SETUP TIMER
HOLDOFF         RMB     1               ; TRANSMISSION HOLDOFF
XPEND           RMB     1               ; XMIT PENDING
MISSED          RMB     1               ; MISSED IR RECEPTION ON
                                        ; LAST CYCLE FLA
MULBLOCK        RMB     1               ; MULTIPLYER FOR BLOCKED TIMER
SERVICE         RMB     1               ; SERVICE
QUIETMO         RMB     1               ; QUIET MODE FLAG, 1=NO
                                        ; TRANSMISSIONS
TOTMSEC         RMB     1               ; TOTAL MISSED SECONDS COUNTER
LVALID          RMB     1               ; LAST PULSE WAS A VALID PULSE
ALLRSYN         RMB     1               ; ALLOW-RESYNC-ATTEMPT FLAG
LOOKTIME        RMB     1               ; LOOK FOR PULSE TIMER
REPEAT          RMB     1               ; REPEAT ON NEXT SECOND

;;;;;;; THIS IS THE START OF THE TRANSMITT DATA PACKET
TYPEID          RMB     1               ; TYPE ID FIELD
SHORTSN         RMB     4               ; SHORT SERIAL NUMBER THIRTY BITS
CYCOPEN         RMB     3               ; COUNTER FOR OPEN SWITCH
TOTMISS         RMB     3               ; COUNTER FOR OUT OF SYNC
WWCRC           RMB     2               ; CRC16
;;;;;;; THIS IS THE END OF THE TRANSMITT DATA PACKET

WWBYCT          RMB     1               ; BYTE COUNT
TTIME           RMB     1               ; TRANSMITT CELL TIME
NODE            RMB     1               ;
RANDOML         RMB     1               ; RANDON COUNTER
RETRYX          RMB     1               ; 10 SECOND RETRY TRANSMITTER

      org     EPROM                     ; start at the top of EPROM

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
```

```
SERIAL      FCB  $30,$10,$20,$00  ; SERIAL NUMBER

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

POWER       RSP
            bclr  irqe,iscr          ; disable external int
            lda   #%00100001         ; ON THE TSOP700
            sta   PORTA              ;
            lda   #%11100101         ;[2]
            sta   ddra               ;[4]
            lda   #%11111111         ;[2] DISABLE PULL DOWNS.
            sta   pdra               ;[4]
            bset  irqr,iscr          ;[3] clear the irq
            lda   #$00               ;[2] LOW OUTPUTS
            sta   PORTB              ;[4]
            lda   #$08               ;[2] PB2 IS AN INPUT
            sta   ddrb               ;[4]
            lda   #$00               ;[2] Enable pulldowns ON PORTB
            sta   pdrb               ;[4]
            TST   QUIETMO            ;[4] TEST FOR QUIET MODE
            BNE   POWER1             ;[3]
            LDA   LVALID             ;[3]
            CLR   LVALID             ;[4] LAST PULSE INVALID BY DEFAULT
            TSTA                     ;[3]
            BNE   PULSE              ;[3] YES, PULSE ANOTHER
            TST   ALLRSYN            ;[4] RESYNC ATTEMPT ALLOWED
            BNE   RESYNC             ;[3] 53 CYC
            JSR   MTOTAL             ; TOTAL THE MISS
            BRA   SLEEP
POWER1      DEC   QUIETMO            ; GET THE FLAG/TIMER
            BNE   SLEEP              ; BUMP IT TOWARD ZERO
            LDA   #TPS               ; RESET THE TIMER/FLAG
            STA   QUIETMO            ; FOR 1 SECOND LCD TICK
            BCLR  LCDCTR,PORTA       ; COUNT THE TICK
            BSET  LCDCTR,DDRA        ; ON THE LCD
            BRA   SLEEP

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

RESYNC      LDA   #$08               ;[2] + 53-cyc sense ir sense on
RESYNC1     DECA                     ;[3] MUST NOT TEST FOR 50MS,
                                     ;    100 CYC
            BNE   RESYNC1            ;[3] 53+N*6=100, N=8

RESYNC2     LDA   #$FF               ;[2]
            STA   LOOKTIME           ;[4] CLEAR THE OVERALL LOOK TIMER
RESYNC3     BRCLR IRIN,PORTA,IRVT    ;[5] HIT OR TIMEOUT!
            DEC   LOOKTIME           ;[5] MOVE THE LOOK TIMER
            BNE   RESYNC3            ;[3] (7.5)*256=1920 CYC, 1 MILLI

            JSR   NOWOP              ; BEAM PATH IS NOW OPEN
            JSR   MTOTAL             ; TOTAL THE MISS
            JSR   CYCEND             ; DO THE CYCLE END STUFF
            TST   SETUPMOD           ; RESET THE SETUP TIMER
```

```
            BEQ   RESYNC4           ; IF IN SETUP MODE
            LDA   #$FF
            STA   SETUPMOD
RESYNC4     BRSET IRIN,PORTA,RESYNC4; HIT OR TIMEOUT!
            BRA   IRVT

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

PULSE       LDA   #$09              ;[2] 2+ 46-cyc sense ir sense on
PULSE1      DECA                    ;[3] MUST NOT TEST FOR 50MS, 100
                                    ;    CYC
            BNE   PULSE1            ;[3] 48+N*6=100, N=9

            LDA   #$07              ;[3] THEN LOOK FOR LOW WITHIN 50
                                    ;    MS.
PULSEX      BRCLR IRIN,PORTA,IRVT   ;[5] POSSIBLE HIT OR TIMEOUT!
            INC   RANDOML           ;[5] RANDOMIZE
            DECA                    ;[3]
            BNE   PULSEX            ;[3] 16*N=100, N=7
            BRA   INVALID

IRVT        LDA   #$09              ;[2] MUST GO HIGH AGAIN WITHIN 50MS
IRVTL       BRSET IRIN,PORTA,IRVALID;[5] AFTER GOING LOW
            DECA                    ;[3]
            BNE   IRVTL             ;[3] 2+11*N=100CYC, N=9
INVALID     BRA   RESYNC2           ; ATTEMPT TO RESYNC.

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

IRVALID     BSET  RBASE,PORTA      ; RESET TIMEBASE, NOW IS 0 TIME
            BCLR  IRPWR,DDRA       ; POWER DOWN THE SENSOR
IRHIT       TST   MISSED           ; THE PATH IS CLOSED ONLY IF
            BNE   IRHIT2           ; THE LAST PULSE WAS ALSO VALID
            TST   SETUPMOD         ; TEST MODE?
            BEQ   IRHIT1
            BSET  LED,PORTA        ; FLASH THE LED IN TEST MODE
            BSET  LED,DDRA
            DEC   SETUPMOD         ; COUNT THE FLASH
IRHIT1      JSR   NOWCL            ; BEAM PATH IS NOW CLOSED
IRHIT2      CLR   MISSED           ; CLEAR THE MISS CTR
            CLR   MULBLOCK         ; AND THE MULTIPLYER FOR BLOCKED
            LDA   #1               ; SET PULSE VALID
            STA   LVALID
            BRA   SLEEP

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

SLEEP       BSR   CYCEND            ; DO THE CYCLE END STUFF
            BCLR  RADIO,PORTA      ; RADIO OFF
            BCLR  IRPWR,DDRA       ; ir sub system off
            BSET  irqr,iscr        ; clear the irq
            STOP

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
```

```
CYCEND      LDA   SECOND          ; SUB SECOND COUNTER
            BEQ   CYCENDRL        ; EXPIRED SECOND, RELOAD
            DECA
            STA   SECOND
            TST   SERVICE         ; TRANSMIT SERVICE NOW
            BNE   CYCENDXM
            TST   SECOND
            BNE   CYCENDS

            TST   SETUPTIM        ; SETUP TIMER RUNNING?
            BEQ   CYCENDRL
            DEC   SETUPTIM
            BNE   CYCENDRL
            CLR   SETUPMOD        ; EXIT SETUP MODE, TIMEOUT
CYCENDRL    LDA   #TPS            ; SET  TICKS/SEC
            STA   SECOND
            LDA   SUBCTR          ; SUB INTERVAL COUNTER
            DECA
            AND   #$3F
            STA   SUBCTR
            BNE   CYCEND1
            LDA   RANDOML         ; RANDOMIZE
            LDX   RANDOML
            EOR   POWER,X
            AND   #$07
            ADD   #XMPERIOD       ; ADD LSBIT TO TRANSMITT PERIOD
            STA   SUBCTR          ; RANDOM 0-7 SECOND SLIP OF
                                  ; INTERVAL
            STA   XPEND
CYCEND1     TST   SERVICE         ; TRANSMIT SERVICE NOW
            BNE   CYCENDXM
            TST   REPEAT          ; REPEAT PACKET?
            BEQ   CYCEND2
            CLR   REPEAT          ; KILL REPEAT FLAG
            BRA   CYCENDXM        ; TRANSMIT AGAIN
CYCEND2     TST   HOLDOFF         ; HOLDOFF TIMER
            BEQ   CYCEND3
            DEC   HOLDOFF
            BNE   CYCENDS
CYCEND3     LDA   XPEND           ; XMIT PENDING?
            BEQ   CYCENDS         ; NO.
            STA   REPEAT          ; SET TO REPEAT
CYCENDXM    TST   QUIETMO         ; DO NOT TRANSMITT IN QUIET MODE
            BNE   CYCENDNX
            JSR   XPACKET         ; TRANSMIT PACKET
CYCENDNX    CLR   XPEND           ; CLEAR PENDING FLAG
            LDA   #$0A            ; HOLD OFF FOR 10 SECONDS
            STA   HOLDOFF         ; THE NEXT TRANSMISSION
            TST   RETRYX          ; RETRYS REMAINING?
            BEQ   CYCENDS
            DEC   RETRYX
            LDA   #$0A            ; SET 10 SECOND RETRY XMIT TIMER
            STA   SUBCTR
CYCENDS     JSR   PROCPB          ; PROCESS THE PUSH BUTTON
```

```
CYCENDSS     RTS

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

NOWCL        BCLR  OSWITCH,STATUS      ; "CLOSED", CLEAR THE OPEN STATUS
             BRSET CSWITCH,STATUS,NOWOPE  ; IS IT NEW?
             BSET  CSWITCH,STATUS      ; "NEW" SET THE CLOSED STATUS
             BRA   XPRETRY             ; CAUSE A TRANSMISSION
;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;


NOWOP        BCLR  CSWITCH,STATUS      ; "OPEN", CLEAR THE CLOSED STATUS
             BRSET OSWITCH,STATUS,NOWOPE
             BSET  OSWITCH,STATUS      ; NEW, SET THE "OPEN" STATUS
             BCLR  LCDCTR,PORTA        ; COUNT THE HIT
             BSET  LCDCTR,DDRA         ;
             INC   CYCOPEN             ; COUNT THE ACTIVE STATE
             BNE   XPSETLED            ; TRANSMIT DATA
             INC   CYCOPEN+1
             BNE   XPSETLED            ; TRANSMIT DATA
             INC   CYCOPEN+2
XPSETLED     BSET  LED,PORTA           ; FLASH THE LED
             BSET  LED,DDRA            ;
XPRETRY      LDA   #3                  ; SEND THREE MORE AT 10 SEC PERIOD
             STA   RETRYX
             LDA   #1                  ; SET THE XMIT PENDING FLAG
             STA   XPEND               ; TO ENABLE TRANSMIT DATA
NOWOPE       RTS

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

MTOTAL       INC   TOTMSEC             ; SUB SECOND COUNTER
             LDA   TOTMSEC
             CMP   #TPS                ; TICKS PER SEDOND
             BLO   MTOTAL1
             CLR   TOTMSEC
             INC   TOTMISS             ; BUMP TOTALS, 8 HRS=007080h
             BNE   MTOTAL1
             INC   TOTMISS+1           ; 256 SEC/COUNT
             BNE   MTOTAL1
             INC   TOTMISS+2           ; 65536 SECONDS/COUNT, 1=18.2 HRS
MTOTAL1      INC   MISSED              ; NOT MAXED, BUMP IT.
             LDA   MISSED              ; GET THE MISSED COUNTER
             CMP   #$E8                ; ACT NORMAL FOR 232-CYC BLOCKED
             BLO   MTOTALM             ; INDICATE WINDOW HERE
             CMP   #$FF                ; (255-232)CYC, REPEAT OPEN WINDOW
             BLO   MTOTALB             ; BLOCK RESYNC ATTEMPT UNTIL WINDOW
             LDA   #$E8                ; YES,SET BACK TO 15 SEC POINT
             STA   MISSED              ; TEST FOR RESYNC EVERY 1.5*4 SEC
             INC   MULBLOCK            ; INCREMENT THE MULTIPLYER
             LDA   MULBLOCK
             CMP   #$04
             BLO   MTOTALB             ; BLOCK RESYNC ATTEMPT ON NEXT CYC
             CLR   MULBLOCK
MTOTALM      LDA   #1                  ; ALLOW RESYNC ATTEMPT ON NEXT CYC
```

```
            STA    ALLRSYN
            RTS
MTOTALB     CLR    ALLRSYN              ; BLOCK RESYNC ON NEXT CYCLE
            RTS

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

PROCPB      BRCLR  PUHBUT,PORTB,PBNOT; IS THE PB PUSHED?
            LDA    PBCTR                ; DO NOT GO ABOVE 255
            CMP    #$FF
            BEQ    POWOOS
            INC    PBCTR                ; BUMP THE PB ACTIVE COUNTER
            LDA    PBCTR                ; IF >4 SECONDS, CLEAR LCD CTR
            CMP    #$40
            BEQ    PROCPB1              ; SET CLEAR LCD/COUNT MODE
POWOOS      RTS

PROCPB1     BCLR   LCDCLR,PORTA         ; SET THE LCD CLEAR LOW
            BSET   LCDCLR,DDRA
            CLR    CYCOPEN
            CLR    CYCOPEN+1
            CLR    CYCOPEN+2
            CLR    TOTMSEC
            CLR    TOTMISS              ; CLEAR THE TOTAL MISSED COUNTER
            CLR    TOTMISS+1
            CLR    TOTMISS+2
            CLR    SETUPMOD             ; CLEAR THE SETUP MODE
            CLR    SETUPTIM
            CLR    QUIETMO              ; EXIT QUIET MODE
            RTS

PBNOT       LDA    PBCTR                ; SET TEST MODE ON PB UP
            CMP    #1                   ; COUNT MUST BE ABOVE 1
            BLS    PBNOTC
            CLR    SETUPMOD             ; CLEAR THE SETUP MODE
            CLR    SETUPTIM
            CMP    #$28                 ; IF LCD CLEAR DO NOT ENTER TEST
            BHS    PBNOTCQ              ; OR IF SERVICE MODE
            LDA    #1                   ; SET SERVICE MODE
            STA    SERVICE
            LDA    #$80                 ; SET TEST MODE TIMER/STATUS
            STA    SETUPMOD             ; HITS REQUIRED TO EXIT SETUP MODE
            LDA    #$3C                 ; 120 SECOND MAX TIME FOR SETUP
            STA    SETUPTIM
            CLR    MISSED               ; CLEAR THE MISS CTR
            CLR    MULBLOCK             ; AND THE MULTIPLYER FOR BLOCKED
PBNOTC      CLR    PBCTR                ; CLEAR THE PB DOWN COUNTER
            RTS
PBNOTCQ     CMP    #$80                 ; 8 SECONDS PUSHED?
            BLO    PBNOTC
            LDA    #TPS                 ; SETUP FOR 1/SEC COUNT
            STA    QUIETMO              ; SET QUIET MODE
            BRA    PBNOTC

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
```

```
XPACKET     LDA   SERIAL            ; GET THE SERIAL NUMBER
            STA   SHORTSN
            LDA   SERIAL+1
            STA   SHORTSN+1
            LDA   SERIAL+2
            STA   SHORTSN+2
            LDA   SERIAL+3
            AND   #$FC
            STA   SHORTSN+3
            LDA   MISSED            ; GET THE MISSED COUNTER
            CMP   #$E8              ; BLOCKED?
            BHS   XPACKBK
            LDA   STATUS            ; GET THE STATUS INPUTS
            AND   #$03              ; MASK THE STATUS
            ORA   SHORTSN+3
            STA   SHORTSN+3

XPACKBK     LDA   #ID               ; SET THE TYPE FIELD
            STA   TYPEID
            TST   SERVICE           ; SERVICE MODE?
            BEQ   RWAVES
            LDA   #SID              ; GET SERVICE ID
            STA   TYPEID
            CLR   SERVICE
RWAVES      BSR   WRCCAL            ; CALCULATE CRC 16 OF DATA
RWAVE       LDA   #$0D              ; SEND BLKSIZE+2 BYTES
            STA   WWBYCT
            LDX   #TYPEID           ; point to ID,data
RWAVEX      BSET  radio,PORTA       ; RADIO ON FOR 1000 MICROSEC
            LDA   #$A7              ; 12*N=2000 CYCLES, N=167
RWAVEI      DECA                    ;[3]
            INCA                    ;[3]
            DECA                    ;[3]
            BNE   RWAVEI            ;[3]
RWAVE1      BSR   Putchar           ;[6]
            INCX                    ;[3]
            DEC   WWBYCT            ;[3] BYTE COUNT IMAGE
            BNE   RWAVE1            ;[3] 15 CYCLES OF OVERHEAD
                                    ; INTRABYTE
            BCLR  RADIO,PORTA       ; RADIO OFF
            RTS                     ;

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

WRCCAL      LDA   #SEEDLSB          ; SEED LSB
            STA     WWCRC+1         ; SEED THE CRC
            LDA   #SEEDMSB          ; SEED MSB
            STA     WWCRC
            LDA     #$0B            ; BYTES TO CRC
            STA     WWBYCT
            LDX     #TYPEID         ; POINT TO DATA RECORD
WRCCAL0     LDA     ,X
            INCX
```

```
WRCBYT      EOR     WWCRC+1         ; EOR DATA WITH CRC LSB
            STA     WWCRC+1
            LDA     #8              ; SET 8 BIT COUNTER
            STA     Node
WRCCAL1     LSR     WWCRC           ; SHIFT IT RIGHT ONE PLACE
            ROR     WWCRC+1
            BCC     WRCCAL2
            LDA     WWCRC
            EOR     #$A0
            STA     WWCRC
            LDA     WWCRC+1
            EOR     #$01
            STA     WWCRC+1
WRCCAL2     DEC     Node
            BNE     WRCCAL1

            DEC     WWBYCT
            BNE     WRCCAL0
            LDA     WWCRC           ; REVERSE LSB/MSB
            COMA
            STA     WWBYCT
            LDA     WWCRC+1
            COMA
            STA     WWCRC
            LDA     WWBYCT
            STA     WWCRC+1
            RTS


;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;;; reset vectors:
;                   org     Timer_INT
;
;                   fdb     Timer_SVR
;
;
;
;                   org     IRQ_INT
;
;                   fdb     IRQ_SVR
;
;
;
;                   org     SWI_INT
;
;                   fdb     SWI_SVR
;
;
;
;                   org     RESET
;
;                   fdb     Start
;
```

```
        org     $07f8
        fdb     POWER
        fdb     POWER
        fdb     POWER
        fdb     POWER

        end
```